
XQuery (2) 

 
Exercise 1: Simple queries in XQuery 
1.1. 
 
<Flights> 
{ 
let $fl := doc("exercise7-1.xml") 
for $a in $fl//Airport[name eq 'North Pole'] 
return $fl//Flight[(source eq $a/@airId) and (date eq '2009-12-24')] 
} 
</Flights> 
 
1.2. 
 
let $doc := doc("exercise7-1.xml") 
let $results := ( 
<order> 
  { 
  for $a in $doc//Airport 
  let $c := count($doc//Flight[((./source eq $a/@airId) 
                               or 
                               (./destination eq $a/@airId)) 
                              ]) 
  order by $c descending, $a/name ascending 
  return 
  <result> 
    {$a} 
    <count>{$c}</count> 
  </result> 
  } 
</order> 
) 
let $maxcount := max($results//count) 
return $results/result[xs:integer(./count) eq $maxcount] 
 
  



1.3. 
 
let $doc := doc("exercise7-1.xml") 
for $p in $doc//Passenger[name eq 'Santa Claus'], 
    $r in $doc//Reservation 
where $r/passRef eq $p/passportnumber 
return 
let $destinations := distinct-values( 
$doc//Airport 
  [@airId eq $doc//Flight 
             [$r/flightRef eq@flightId] 
             /destination 
  ]/name 
) 
return 
<destinations> 
  {for $destination in $destinations 
   return <destination>{$destination}</destination>} 
</destinations> 
 
 
1.4. 
 
let $doc := doc("exercise7-1.xml") 
return 
<Possibilities>{ 
  for $f1 in $doc//Flight[./source eq 'NPL'], 
      $f2 in $doc//Flight[destination eq 'SPL'] 
  where $f1/destination eq $f2/source 
        and xs:time($f1/arrival) lt xs:time($f2/departure) 
        and $f1/seats > 0 and $f2/seats > 0 
  return 
  <OneIntermediateStop> 
    {$f1} 
    {$f2} 
  </OneIntermediateStop>, 
  for $f1 in $doc//Flight[./source eq 'NPL'], 
      $f3 in $doc//Flight[destination eq 'SPL'], 
      $f2 in $doc//Flight 
  where $f1/destination eq $f2/source 
        and $f2/destination eq $f3/source 
        and xs:time($f1/arrival) lt xs:time($f2/departure) 
        and xs:time($f2/arrival) lt xs:time($f3/departure) 
        and $f1/seats > 0 and $f2/seats > 0 and $f3/seats > 0 
  return 
  <TwoIntermediateStops> 
    {$f1} 
    {$f2} 
    {$f3} 
  </TwoIntermediateStops> 
}</Possibilities> 

